



Computing the Jones Polynomial

The function `JP[]` returns the Jones polynomial of a knot - an embedding of S^1 in 3 dimensional Euclidean space defined up to ambient isotopies - while also printing its Kauffman bracket and writhe. The computation of the Kauffman bracket works as the one made in class. For the computation of the writhe to work, the expression of the form $X[i_1, j_1, k_1, l_1] \dots X[i_n, j_n, k_n, l_n]$ representing the knot must be chosen in a systematic way (more so than what is necessary to compute the Kauffman bracket). The method for entering a knot K is as follows :

1. Choose an orientation on K
2. Number the edges of K separated by crossings consecutively following the positive orientation, until coming back to edge 1. Doing so, choose edge 1 so that, at the first crossing encountered in the positive direction along K , the thread going from edge 1 to edge 2 passes 'under' the other thread.
3. For each crossing X in K , multiply the expression for K by the factor $X[i, j, k, l]$, where i is the label of the edge incoming to the crossing on the thread going 'under'. j, k, l are the numbers of the other edges, in the order given by counterclockwise rotation starting from the edge i . That is, $X[i, j, k, l]$ is associated to the crossing drawn below. This process yields the final expression for K .



4. Note that the function `JP` cannot compute the Jones polynomial of the simple unknot (i.e. of a circle without crossings). I suppose it would be easy to deal with this case, but obviously it's not worth the effort... No need of an exoskeleton to compute that.

<< **KnotTheory`**

Loading `KnotTheory`` version of September 6, 2014, 13:37:37.2841.

Read more at <http://katlas.org/wiki/KnotTheory>.

```

Rule[a, b]
a → b

JP[K, Verbose → True]
JP[K, Verbose → True]


Clear[JP]

SetAttributes[P, Orderless];
JP[K_Times, opts___Rule] := Module[{verb, b1, b2, b3, b4, b5, w, J},
  verb = Verbose /. {opts} /. Verbose → False;
  b1 = K //. X[i_, j_, k_, l_] => AP[i, j] P[k, l] + BP[j, k] P[i, l];
  b2 = Expand[b1];
  b3 = b2 //. P[i_, j_] P[j_, k_] => P[i, k];
  b4 = b3 //. {P[i_, j_]^2 → d, P[i_, i_] → d};
  b5 = Expand[b4 //. {B → 1/A, d → -A^2 - 1/A^2}];
  If[verb, Print["The Kauffman bracket is "]];
  If[verb, Print[b5]];
  w = K /. {Times → Plus, X[_ , 1, _ , _] → 1,
    X[_ , _ , _ , 1] → -1, X[_ , j_, _ , l_] => If[j > l, 1, -1]};
  If[verb, Print["The writhe is "]];
  If[verb, Print[w]];
  If[verb, Print["The Jones Polynomial is "]];
  J = Expand[(-A^3)^-w b5];
];
JP[K_PD, opts___] := JP[Times@@K, opts]

```

Examples

Trefoil Knot



```
Ktre1 = X[1, 5, 2, 4] X[3, 1, 4, 6] X[5, 3, 6, 2];
```

```
JP[Ktre1]
```

$$\frac{1}{A^{18}} - \frac{1}{A^{10}} - \frac{1}{A^6} - \frac{1}{A^2}$$

```
JP[PD[Knot[3, 1]]]
```

$$-A^2 - A^6 - A^{10} + A^{18}$$

JP[Ktre1, Verbose → True]

The Kauffman bracket is

$$-\frac{1}{A^9} + \frac{1}{A} + A^3 + A^7$$


The writhe is

3

The Jones Polynomial is

$$\frac{1}{A^{18}} - \frac{1}{A^{10}} - \frac{1}{A^6} - \frac{1}{A^2}$$

Mirror Trefoil Knot



$$\text{Ktre2} = X[1, 4, 2, 5] X[3, 6, 4, 1] X[5, 2, 6, 3];$$


JP[Ktre2]

$$-A^2 - A^6 - A^{10} + A^{18}$$

JP[PD[Knot[3, 1]]]

$$-A^2 - A^6 - A^{10} + A^{18}$$

Knot 8₁₇



$$\text{K817} = X[1, 12, 2, 13] X[3, 8, 4, 9] X[5, 1, 6, 16] X[7, 2, 8, 3] X[9, 15, 10, 14] X[11, 4, 12, 5] X[13, 7, 14, 6] X[15, 11, 16, 10];$$

JP[K817]

$$-\frac{1}{A^{18}} + \frac{2}{A^{14}} - \frac{2}{A^{10}} + \frac{1}{A^6} - \frac{1}{A^2} - A^2 + A^6 - 2A^{10} + 2A^{14} - A^{18}$$

PD[Knot[8, 17]]

KnotTheory: Loading precomputed data in PD4Knots`

```
PD[X[6, 2, 7, 1], X[14, 8, 15, 7], X[8, 3, 9, 4], X[2, 13, 3, 14],
  X[12, 5, 13, 6], X[4, 9, 5, 10], X[16, 12, 1, 11], X[10, 16, 11, 15]]
```

JP[PD[Knot[8, 17]]]

$$1 - \frac{1}{A^{12}} + \frac{2}{A^8} - \frac{2}{A^4} - A^4 - A^8 + A^{12} - 2A^{16} + 2A^{20} - A^{24}$$

AllKnots[{3, 11}] // Length

801

Timing[tab = Table[JP[PD[K]], {K, AllKnots[{3, 11]}]}];]

{460.25, Null}

Union[tab] // Length

739